情報処理 II 資料

Mathematica 入門

桂田 祐史

2005 年 6 月 30 日

この文書 (の訂正版) は

http://www.math.meiji.ac.jp/~mk/syori2-2005/mathematica.pdf

でアクセスできます。

情報処理 II のホームページは http://www.math.meiji.ac.jp/~mk/syori2-2005/

代表的な**数式処理系**である **Mathematica**（マセマティカ） を体験しましょう。数式処理でどういうことが出来るのか大体の雰囲気をつかんで、今後の学習・研究のヒントにしてもらうのがねらい。

1 Mathematica ってこんなもの

以下の例は、数学科の計算機である oyabun にログインして、Mathematica を実行してみたものです (Windows 環境での実行例ではありませんが、本質は同じです)。ここではプログラムなどは書かずに、式を順次入力して計算結果を表示させています。



```
oyabun% math
Mathematica 4.0 for Solaris
Copyright 1988-1999 Wolfram Research, Inc.
 -- Motif graphics initialized --

In[1]:= 1/2 + 1/3                    ← 分数計算

        5
Out[1]= -                            → ちょっと見難いですけどね
        6

In[2]:= a={{0,1},{6,1}}              ← 行列の入力

Out[2]= {{0, 1}, {6, 1}}

In[3]:= Eigenvalues[a]               ← 行列の固有値の計算

Out[3]= {-2, 3}

In[4]:= Eigenvectors[a]              ← 行列の固有ベクトルの計算

Out[4]= {{-1, 2}, {1, 3}}

In[5]:= Expand[(x+y)^6]              ← 式の展開

         6      5         4  2       3  3       2  4        5    6
Out[5]= x  + 6 x  y + 15 x  y  + 20 x  y  + 15 x  y  + 6 x y  + y

In[6]:= N[Pi,50]                     ← 円周率 50 桁

Out[6]= 3.14159265358979323846264338327950288419716939937510

In[7]:= Integrate[Log[x],x]          ← 不定積分

Out[7]= -x + x Log[x]

In[8]:= Plot3D[x^2 - y^2, {x,-1,1}, {y,-1,1}]  ← グラフ

Out[8]= -Graphics-                   → ここで画面に図が表示されます

In[9]:= Solve[x^3+2x==1,x]           ← 3 次方程式を解かせてみる
```

結果は一見に価するけれど、紙を食うのでカットします。

```
In[10]:= ParametricPlot3D[{Cos[t](3+Cos[u]),Sin[t](3+Cos[u]),Sin[u]},
         {t,0,2Pi},{u,0,2Pi}]        ← トーラスを描かせる。

Out[10]:= -Graphics3D-

In[11]:= Quit                        ← 終了
oyabun%
```

2　数式処理とは

プログラミング言語 (計算機言語) の中には、数値や文字だけでなく、

数式をデータとして扱うことの出来る「**数式処理言語**」

と呼ばれるものがあります。数式処理言語を使えるソフトウェアを**数式処理系**と呼びます。現在、一般向けの数式処理系としては **Mathematica**, **Maple**(メイプル) が双璧と言われています。

(その他に **MuPAD**(ミューパッド)[1], **REDUCE**(リデュース)[2], **Risa/Asir**[3], **Macsyma**[4], **MAXIMA**[5] などが有名。)

C のようなプログラミング言語は、プログラムの中では「数式」を書けますが、関数 `scanf()` や `printf()` 等で入出力可能なデータは、数や文字列だけで、例えば `-2/5` のような分数式の入力は出来ません。またグラフを描くプログラムを作る場合に、範囲や、分割数の指定等は実行時に入力出来ても、グラフを描こうとしている関数自体は (普通の方法では) 入力できず、プログラムの中に自分で埋め込むしかなかったわけです。そういう意味では C は不自由な言語であると言えます[6]。

Mathematica は、グラフィックスやサウンドなども便利に扱えるようになっていて、ひょっとすると「数式処理」とだけ説明するのはもう間違いかもしれません。

3　この講義で用いる Mathematica

3.1　Mathematica は商品

C 言語等は国際規格があり、無償で利用できる処理系もありますが、Mathematica は **Wolfram Research** という一企業の所有物で、処理系は同社が作成・販売しているものしかありません。予算の都合上、利用できる個数が限られています。

2005 年度の情報科学センターには、Mathematica 4 のライセンスが 40 (32?) あるそうですので、これを使います。

3.2　数学科学生へ

数学科では Mathematica 5.0 のライセンスを 30 所有しています (同時に 30 台のコンピューターで Mathematica を利用することが出来る)。設定を頑張れば、自宅のパソコンからも利用

[1]個人・非商用利用には無償で利用できるバージョンがあるようです。かなりの完成度なのでチェックしてみると良いかもしれません。

[2]筆者が学生の頃 (20 年前)、大型計算機で REDUCE を使って、計算するのがおしゃれだった。現在でも計算の種類によっては、一番かもしれない。

[3]Made in Japan の現役。グレブナー基底の計算など得意です。

[4]かつて MIT でしか使えなかった憧れの (歴史的) 処理系。古い本を読むと良く出て来ます。

[5]Macsyma の子孫。GPL (GNU GENERAL PUBLIC LICENSE) で配布されている (ゆえに、いわゆるフリーソフト)。メジャーになれるか ??

[6]もちろん不自由さを補って余りある大きな利点があるから、現在でも盛んに使われているわけです。例えば、実際の処理系の (反復の多い) 数値計算の速さで比べると C が圧勝します。原理的には一つのプログラミング言語があれば、どんな計算でも出来るはずなのですが、実際的な意味で万能のプログラミング言語と呼べるものは存在せず、適材適所を心がけることが重要です。みなさんも、あまり一つの言語、一つのシステムにこだわらずに、機会があったら色々なものを勉強してみましょう。

できるようになります (ただし明治大学数学科の学生である期間に限ります)。使いたい人は要相談。

4 基本的な使い方 (Windows 向け説明)

4.1 起動

情報処理教室の Windows XP パソコンでの Mathematica の利用法

1. スタート・ボタンをクリック
2. [すべてのプログラム (P)] → [MATHEMATICA] を選択
3. User ID (ユーザ名) とパスワードを尋ねられたらきちんと答える。
4. タイトルバーに `Z!Stream XPStyle - Microsoft Internet Explorer` と書かれたウィンドウが現れる。[Start] → [プログラム] → [Mathematica 4] を選択。

画面左側に現われる “Untitled-1” (日本語版では「名称未定義-1*」) というウィンドウに、キーボードからコマンドを入力して、最後に Shift+Enter を入力するのが基本である。(コマンドの入力に適宜、テンプレート・キーを用いると便利。)

4.2 起動後の基本操作

- コマンドをキーボードから入力してから、最後に Shift+Enter をタイプする。
  コマンドの部分に “`In[番号]:=` ” というラベルが加えられ、“`Out[番号]=` ” に続いて計算結果が出る。

`In[番号]:=コマンド` を**入力セル**、`Out[番号]=計算結果` を**出力セル**と呼ぶ。

- 一度評価 (計算) した入力セルを書き換えて再び Shift+Enter を入力すると、再評価して、その結果で出力セルを上書きする。
- コマンド入力の最後にセミコロン (;) を書くと、計算結果は表示されない[7]。
- 入力に関して便利な記号として
  - `%`　直前の結果
  - `%%%`$\cdots$`%` (k 個重ねる)　k 回前の結果
  - `%`n (n は自然数)　Out[n] と同じ (n 番目の結果)。

[7]意味がないように思うかも知れないが、`a=2^1000` のような変数への代入など、副作用のある処理をする場合に使うことがある。

4.3 計算を強制的に止める

時々計算が暴走することがあります (決して終わらない計算を実行するなど)。そういう場合など、現在の計算を止めさせたくなったときは、[Kernel] メニューから [Abort Evaluation] を選択します。計算に夢中だとすぐには止まってくれませんが。

4.4 計算結果の保存 (ノートブックの利用)

現在の Mathematica では、入力されたコマンドと計算結果を、**ノートブック**という形式で保存することが出来る。そのためには [File] メニューから [Save As] を選んで、ファイルを置くフォルダとファイル名を指定する。次の二点に注意。

1. 現在の情報処理教育用パソコンの Windows の設定では、ユーザーのファイルは、マイドキュメント・フォルダの下に置くべきである (Mathmetica のシステム・ディレクトリィや、デスクトップに置いてはいけない[8])。

2. ファイル名は "`Untitled-1.nb`" から変更するのが無難。ファイル名から中身が推測できるようなものが望ましい。拡張子は `.nb` とすること。

ファイルの名前は衝突しないように気をつけよう。くれぐれも一つのファイルを二つのプロセスで扱うようにしたりしないこと。某学生がノートブックの内容を壊してしまいましたが、その理由はこれであると推察されます (後で追試したら同じエラーメッセージを再現できたので多分当たっていると思います)。

レポート提出用のノートブックを作る

レポート提出用のノートブックを作るには、例えば次のようにするとよい。

1. 課題をこなすための計算を一通り実行した後で、[File] メニューから [New] を選んで新しいノートブックを開き、そちらに必要なコマンドを Copy, Paste して、再実行することで、必要なところだけを抜き出したノートブックを作る。

File メニューから [Save As] を選び、マイドキュメントに適切な名前 ("syori20630" とか "課題 6" とか) で保存する (拡張子は .nb となる)。

保存してあるノートブックを開く

情報処理教室の環境では、Mathematica のノートブックを Windows に登録していないので、ノートブックのアイコンをクリックしても Open することはできない。Mathematica の File メニューから Open コマンドを選んでファイルを選択することで開ける。

[8] デスクトップに置いたファイルは、ログオンしている間だけしか、保存されない。

4.5 パレット

Mathematica を起動すると、次のような入力を補助するためのウィンドウ (**パレット**) が現れる。ときどきこのウィンドウが消えたと騒ぐ人が多いが、[File] → [Palettes] → [4 BasicInput] で再表示できる。

4.6 カレント・ディレクトリィ

カレント・ディレクトリィを知るには `Directory[]` を実行する。
カレント・ディレクトリィを変えるには `SetDirectory[]`

```
SetDirectory["Z:\\"]
```

5 電卓的な使用

最初の「Mathematica ってこんなもの」を試してみた後は、簡単な文法と、関数を知るだけでかなり便利に使えるはずです。

5.1 かっこ

- ( ) は結合順位をグループ化する普通のかっこ。
- [ ] は関数 (コマンド) に渡す引数を示す。
- { } はリストを作るかっこ (リストについては後述)。
- [[ ]] はリストの要素のアクセス (これも後述)。

5.2 変数 (名前) の宣言、変数への代入

- 名前では大文字小文字を区別します。組み込みの名前は先頭が大文字になっています。ユーザーはなるべく小文字を使うことを奨励されています (衝突を回避するため) 。
- a=式, a=b=式, 等で代入。
- ?Global‘* とすると、(自分が) 現在何を定義しているか表示します。
- ?*Sin* とすると、“Sin” を含む名前の一覧が表示されます。
- a=. または Clear[a] で変数 a の内容を除きます (内容を除く — 名前は残ります)。Clear[a,b,···] のように複数の変数の内容を同時に除くこともできます。
- Remove[a,b,···] 名前まで込めて完璧に除く。

どういう名前が定義されているかは、??Global‘* でチェックできます。全部削除するには Remove["Global‘*"] とすると良いでしょう。

不要になった変数 (または後で出て来るユーザー定義の関数) は、Clear[名前] または Remove[名前] 等でこまめに削除しておくことを勧めます (一度使った変数に残っているゴミのせいで首をひねることが良くあります)。

変数の定義は不要になったら消さないと副作用が出る

```
In[1]:= f=x^2+2x+3

                     2
Out[1]= 3 + 2 x + x

In[2]:= x=1

Out[2]= 1

In[3]:= f

Out[3]= 6

In[4]:= D[f,x]

General::ivar: 1 is not a valid variable.

Out[4]= D[6, 1]

In[5]:= Clear[x]

In[6]:= D[f,x]

Out[6]= 2 + 2 x
```

5.3 演算子

- 四則演算は普通の記号で OK。a+b, a-b, a*b または a b (空白抜きで ab は一つの名前。a5 も一つの名前だが、5a は 5 かける a の意味), a/b.
- べき乗 a^b, 階乗 a! もあります。

5.4 組み込み関数

Sqrt[x], Exp[x], Log[x], Log[b,x], Sin[x], Cos[x], Tan[x], ArcSin[x], ArcCos[x], ArcTan[x], Abs[x], Round[x], Random[], Max[x,y,...,z], Min[x,y,...,z] のように普通のプログラミング言語にそなわっているような関数はもちろん、色々なものがあります (いざとなったら Help を見て下さい)。

以下、重要なことを (重複をいとわず) いくつか紹介します。

- 関数の引数を囲む括弧は [ ]

- システム組み込みのものは名前の先頭の文字が大文字です。
- ?文字列* で一覧表、さらに ?関数名 や ??関数名 でその関数の説明が出ますが、Help から探す方が良いかもしれません。
- 関数がどういうオプションを持っているか見るに Options[] を用います。

Options[Plot]	Plot のオプション全部を表示
Options[Plot, PlotRange]	Plot の PlotRange オプションを表示

- オプションを一時的に変更するには “SetOptions[]” を用います。

5.5 無限多倍長整数 (小数も任意精度まで)

整数や有理数等は (可能な限り) 誤差のない計算をします。小数に対しては有限精度の計算となります。その場合 Sin[] などの関数の精度はデフォールトでは C 言語の double と同程度 (つまりパソコンや WS では 10 進 16 桁弱) ですが、精度はかなり自由に指定できます。

整数や有理数については、特に意識せずに使うことが出来ます。100! や 2^100 などで試せます。

5.6 分数

1/2+1/3 のようにして分数が入力できて、分数のまま計算できます。

5.7 ルートなど

ルートなどのシンボリックな数も使えます。例えば Expand[(-1+Sqrt[3]I)^3] など。

5.8 有限桁の小数への変換 (近似値の計算)

分数や Sqrt[3] のようなシンボリックな数を (有限桁の) 小数に変換するのに、 N[] が使えます: N[(1+Sqrt[5])/2]

特に N[Sqrt[2], 100] のように精度を指定できます。

```
(1+Sqrt[3])^2; N[%,100]   直前の結果を 100 桁
N[Pi,100]
N[E,100]
2^100 // N
N[Sin[30 Degree]]          sin 30° の近似値
```

5.9 文字式 (多項式、分数式、その他)

```
2 f[x] + 3 f[x]
p1 = Expand[(1+x)^10]          多項式の展開結果を変数に代入
p2 = (1+x)^3
PolynomialQuotient[p1,p2,x]    多項式の商
PolynomialRemainder[p1,p2,x]   多項式の剰余
PolynomialGCD[p1,p2]           最大公約多項式
Apart[1/(x^3-1)]               部分分数への分解
Remove[p1,p1]                  おそうじ
```

5.10 定数

Pi(円周率 π), E(自然対数の底 e), Degree($=180/\pi$), I(虚数単位 $i=\sqrt{-1}$), Infinity (無限大 $+\infty$), ComplexInfinity (複素平面の無限大), GoldenRation (黄金比) などの定数があらかじめ定義されています。

5.11 整数 (素因数分解、素数判定)

```
n=2^2^5+1          n = 2^2^5 + 1 = 4294967297
PrimeQ[n]          n は素数かどうか判定する
FactorInteger[n]   n の素因数分解
Remove[n]          おそうじ
Mod[123456,123]    123456 を 123 で割った余り
GCD[96,18]         最大公約数
```

5.12 複素数

```
z=(3 + 4I) (1 + 2I)          (3 + 4i)(1 + 2i)
Re[z]                        実部
Im[z]                        虚部
Conjugate[z]                 共役複素数
Abs[z]                       絶対値
Arg[z]                       偏角
ComplexExpand[E^(Pi I/6)]    a + ib の形にする
Remove[z]                    おそうじ
```

5.13 結果の簡単化

計算結果が自分の望んでいる形に表されないことがしばしばあります。まず Simplify[] という関数を覚えておきましょう。結果が複雑だったら、取りあえず Simplify[%] として、「直前の結果の簡単化」を試みるのが良いでしょう (変わらないことも多いですが)。ちなみに、通分は Together[], 因数分解は Factor[], 展開は Expand[] です。例えば

```
y=1/(1+x)+1/(1-x)
Together[y]
Simplify[y]
Factor[y]
Remove[x,y]
```

その他に FullSimplify[] などがありますが、少し時間がかかります。

5.14 一時的な代入

式に含まれる変数に一時的に値を代入して式の値を求めたい場合は、

式 /. 名前 -> 値

とします。具体的には、例えば

```
y = x^2
y /.  x->1
y                y 自体は変っていない
Remove[x,y]
```

一度に複数の代入をするには、括弧 { } でくくって、

式 /. {名前 -> 値, 名前 -> 値,...}

とします。

```
(x + y + z + w)^2
% /.  {y->1,z->2}  y に 1 を、z に 2 を代入
Remove[x,y,z,w]
```

方程式の解を Solve[] (詳しいことは後述) を用いて、求めたときの結果はこの代入をするのに便利です。例えば $x^2+x+1=0$ の根の 3 乗を計算するには、次のようにすれば OK.

```
Solve[x^2+x+1==0,x]
x^3 /.  %              解の 3 乗を計算してみる
Remove[x]
```

5.15 微分

`D[x^n,x]`	$(x^n)'$
`f=x^2+2x y+3y^2+4x-5y+6`	
`D[f,x]`	f_x
`D[f,x,y]`	f_{xy}
`D[f,{x,2}]`	f_{xx}
`D[f,{x,2},{y,3}]`	f_{xxyyy}
`Remove[f]`	
`f[x_]:= x^2+2x+3`	
`f''[x]`	
`Remove[f]`	

5.16 不定積分、定積分

`Integrate[式、変数]`, `Integrate[式、変数の範囲]` とします。数値積分版 `NIntegrate[]` もあります。後者は近似値しか計算できませんが、前者では計算できないような定積分も扱えます。

```
Integrate[1/(1+x^2),x]
Integrate[x^2,{x,0,1}]
Integrate[E^(-x^2),{x,0,Infinity}]
NIntegrate[Sqrt[Sin[x]], {x,1,2}]
NIntegrate[Sqrt[Sin[x]], {x,1,2}, WorkingPrecision->50,AccuracyGoal->40]
```

5.17 級数 (シグマ, Taylor 展開)

`Sum[n,{n,1,5}]`	$\sum_{n=1}^{5} n$
`Sum[k^2,{k,n}]`	$\sum_{k=1}^{n} k^2$
`Series[Exp[x],{x,0,10}]`	$x=0$ のまわりの 10 次までの Taylor 展開！

5.18 方程式を解く

`Solve[]`, `NSolve[]` という二つの手続きがあります。
`Solve[左辺==右辺, 未知数]` は式変形で解きます。

```
Solve[x^2+3x+2==0, x]
```

結果は {x->1} のような規則のリストの形で得られますが、x /. % とすれば、解のリストになります。

次のような連立方程式も解けます。

```
Solve[{x+y+z==6, 2x-y+z==5,-3x+y+2z==0},{x,y,z}]
```

Mathematica は 2 次方程式だけでなく、3 次方程式、4 次方程式の根の公式も覚えていて解くことが出来ますが、やってみれば分かるように、結果は分かりにくくなることが多いです。どの程度の値なのか知りたい場合、つまり近似値でよければ直後に % // N あるいは N[%, 50] のように入力すれば求められます。

```
Solve[x^3+2x^2+3x+4==0, x]
% // N                          直前の結果を小数で
N[%%, 50]                      二つ前の結果を 50 桁の小数で
```

文字を係数に含む方程式も解くことが出来ます。

```
Solve[a x + b ==0, x]
Reduce[a x + b ==0, x]  ちゃんと場合わけをする
```

もっとも、特別簡単なものを除けば、方程式は式変形のみで解くことは難しいです (できないことが多い)。そういう場合は、近似値を求めることで我慢することにすれば、NSolve[], FindRoot[] などの関数が利用できます。

```
NSolve[x^3+2x^2+3x+4==0, x, 40]      精度 40 桁で解く
FindRoot[x^3+2x^2+3x+4==0, {x, 0}]   Newton 法で 0 の近くの解を探す。
```

5.19 極限

その名もずばり Limit[] という関数があります。その際 ∞ を意味する “Infinity” が使えることに注目。

```
Limit[Sin[x] / x, x-> 0]
Limit[(x^2 + 2 x + 3)/(3 x^2 + 2 x + 1), x->Infinity]
```

いわゆる片側極限 $\lim_{x\uparrow a} f(x), \lim_{x\downarrow a} f(x)$ も計算できます。

```
Limit[Tan[x], x-> Pi/2, Direction -> 1]    下から (左から) の極限
Limit[Tan[x], x-> Pi/2, Direction -> -1]   上から (右から) の極限
```

5.20 リストとその応用 (行列、ベクトル)

いわゆる中括弧 { と } の中にカンマで区切って複数のものを並べて作った「リスト」というデータ構造があります。

`list = {1,2,3}` 1, 2, 3 という 3 つの要素からなるリスト `list` を定義

- 関数の多くはリストに対して作用します。

```
Log[{a,b,c}]
N[{1/2,1/3,1/4}]
Sin[Pi/{1,2,3,4,5,6,7,8}]
```

- 結果をリストとして返す関数は多いので (方程式の解が複数個ある場合とか、固有値、固有ベクトルを求める関数など)、リストの中から特定の要素を取り出す `Part[番号]` や `[[番号]]` は重要です。

```
list = {1,2,3}
Part[list,2] または list[[2]]                    list の第 2 要素
Eigenvalues[{{1,2},{3,4}}]                       行列の固有値を計算
lambda1 = %[[1]]; lambda2 = %[[2]]               それぞれ変数に代入
{lambda1,lambda2}=Eigenvalues[{{1,2},{3,4}}]     同時に代入
a={{1,2},{3,4}}
Part[a,1] または a[[1]]                          a の第 1 行
Part[a,1,2] または a[[1]][[2]]                   a の (1, 2) 成分
```

- リストは集合、ベクトル、行列、テンソル、積分やプロットなどの範囲指定、関数の引数のグループなどを表すのに使えます。

- 集合的な演算をするための関数として `Union[]`, `Intersection[]`, `Complement[]` などがあります。

- ベクトル的な利用法として

```
{1,2,3}+{a,b,c}
2 {1 3 5}
{1,3,5} / 3
{1,2,3} . {3,4,5}   内積
{1,2,3} {2,3,4}     成分ごとの積
{1,2,3} / {2,3,4}   成分ごとの商
```

- 行列演算もできます。

```
A={{a11,a12,a13},{a21,a22,a23},{a31,a32,a33}}
y={y1,y2,y3}
A . y                                          行列とベクトルの積
                                               (ドットが必要なことに注意)
```

行列用の関数としては、行列式 Det[], 転置行列 Transpose[], 逆行列 Inverse[], 固有値 Eigenvalues[], 固有ベクトル Eigenvectors[] などがあります。

- Lisp 言語風の関数として First[], Rest[] などがあります。
- その他 Length[{a,b,c,d,e}], MemberQ[{a,b,c,d,e,f},a], Count[{a,b,a,b,a,b},a], Reverse[], Sort[], RotateLeft[], RotateRight[] 等々。
- リストを生成する Table[] という関数も慣れると便利です。

```
Table[i^2, {i,6}]
Table[Sin[n Pi/5], {n,0,4}]
Table[x^i+2i, {i,5}]
Table[Sqrt[x], {x, 0, 1, 0.25}]
Table[x^i+y^j, {i,3}, {j,2}]
Table[PrimeQ[2^2^p+1],p,8]
Table[{2^2^n+1,PrimeQ[2^2^n+1]},{n,6}]
```

- ReadList[] という関数を使って外部ファイルからリストに読み込める。

```
ReadList["ファイル名", Number]
ReadList["ファイル名", Number, RecordList]
ReadList["外部コマンド名", Number]
ReadList["!外部コマンド名", Number, RecordList]
```

“square.data” を

```
1 1
2 4
3 9
4 16
```

という内容のファイルとする時、以下のコマンドで何が起こるか？

```
ReadList["square.data", Number]
ReadList["square.data", Number, RecordLists -> True]
```

6 プログラミング機能

Mathematica は、これまでに述べてきたような電卓的な使用法、つまり一つ一つの計算の指示を人間が手で入力する仕方でも十分役立つが、通常のプログラミング言語にひけを取らないプログラミングの機能を持っている。これまでに説明していないもので、プログラミングに必須 or 役立つものをあげてみよう。

- 複数の文を並べること
- 条件判断 (真偽の判定)
- 条件判断に基づく分岐
- 繰り返し処理用の専用メカニズム
- まとまった処理に名前をつけて一つの単位とする

6.1 関数定義

(関数の話については、http://www.math.meiji.ac.jp/~mk/syori2-2005/jouhousyori2-2005-10/ の中に新しい説明を書いた。そのうちこの文書にマージする予定である。)

Mathematica のプログラムは (ユーザー定義の) 関数の集合という形になる。

百聞は一見にしかず。$f(x) = x^2$ なる関数 f は、

```
f[x_] := x^2
```

で定義出来る。“関数名 [引数名_] := 式” という形式で定義する。(また “f[名前_ 型名] := 式” のように型名を指定することも出来る。) 関数を再定義すると古い定義は消えてしまう。

```
f[4]
f[a+1]
f[3x+x^2]
```

もちろん関数 “f” の定義が見たい場合は、“?f”, または “??f” とする。

多変数の関数も定義できる。

```
f[x_,y_] := (x^2 - y^2) / (x^2 + y^2)
```

変数の特定の値に対する関数値を代入文で定義することも出来る。次の例では Fibonacci 数列を計算する関数 f[] を定義している。再帰的な定義が出来ることに注意しよう。

```
f[x_] := f[x] = f[x-1]+f[x-2] ; f[1]=1; f[2]=1
Table[f[n],{n,10}]
```

他の例として

```
f[x_] := Sin[x]/x; f[0]=1
Plot[f[x],{x,-2Pi,2Pi}]
```

6.2 基本的なプログラミング機能、特に制御構造

6.2.1 複数の文を並べる

(1) ";" で区切って並べることができる。

(2) “(”, “)” で括ることもできる。

6.2.2 条件判断 (論理式)

関係演算子、論理演算子は C 言語のそれに良く似ている:

`a == b`	等しいか？
`a != b`	等しくないか？
`a < b`	a は b より小さいか？
`a > b`	a は b より大きいか？
`a <= b`	a は b より小さいか、等しい？
`a >= b`	a は b より大きいか？
`&&`	「かつ」.
`||`	「または」.
`!`	「否定」.

論理式の値は、False = 偽, True= 真, である。次の例を試してみること。

```
1+1 == 2
2^3 == 7
1 < 2 < 3
2 != 3
LogicalExpand[(p || q) && !(r || s)]
```

6.2.3 条件判断による分岐

If[*test*, *then-statement*, *else-statement*] とすると分岐が実現できる。例えば

```
If [1+1==2, Print["Yes, you are right."], Print["No, you are wrong."]]
```

6.2.4 繰り返し構文

計算機のプログラムで広い意味の繰り返しは重要である。それを実現するために、再帰的関数 (手続き) 定義や、繰り返しの構文が用意されている。

Do 関数 使い方は "`Do[`*statement*, *iterator*`]`". Fortran の do 文、C の for 文、BASIC の FOR NEXT 構文、Pascal の for 文に似ている。*iterator* (繰り返し指定) で指定しただけ *statement* (文) を繰り返して実行する。

```
Do[Print[i!], {i,5}]              i=1,2,3,4,5
Do[Print[2^i], {i,0,5}]           i=0,1,2,3,4,5
Do[Print[I^i], {i,0,10,3}]        i=0,3,6,9
r=1; Do[r=1/(1+r), {100}]; r      100 回
Do[Print[i], {i,2a,4a,a}]         i=2a,3a,4a
Do[Print[r], {r,0.0,3.5}]         r=0.0,1.0,2.0,3.0
Do[Print[{i,j}], {i,3}, {j,i}]    do i=1,3
                                    do j=1,i
                                      Print {i,j}
                                    end do
                                  end do
```

While 関数 使い方は "`While[`*test*, *statement*`]`". C や Pascal の while 文に似ている。*test* が真である間だけ *statement* を繰り返す。

```
i=1; While[i <= 10, Print[i]; i++]
```

この例では "`i <= 10`" が *test*, "`Print[i]; i++`" が *statement* になっている。

For 関数 使い方は "`For[`*statement1*, *test*, *statement2*, *statement*`]`". C の for 文に似ている。

```
For[i=1, i <=10, i++, Print[i, " ", i^2]]
```

繰り返しの指定 Do 文で使われている iterator は、あちこちで使われる。和 $\sum$ を計算する `Sum` や乗積 Π を計算する `Product`、表を作る `Table` など。

`Sum[i, {i,1,10}]`	$i = 1, 2, \cdots, 10$
`Product[x-i, {i,0,5}]`	$i = 0, 1, \cdots, 5$
`Product[e, {e,x,x-5,-1}]`	$e = x, x-1, x-2, x-3, x-4, x-5$
`Table[i!, {i,5}]`	$i = 1, 2, 3, 4, 5$

`Print[]` の結果は表示されるだけで、後には残らないので、残して解析したい場合は `Table[]` などを使うのがお勧めである (このあたりは C 言語のプログラミングの発想とは異なる)。

6.3 関数定義のための細かい注意

以下の記述はプログラミングの中級者向けであって、最初は読み飛ばして構わない。実用的なプログラムを書く際に必要な、他への悪影響が出ない&他からの影響を受けないようにするための工夫の話。

局所変数の利用 特別なことをしないかぎり、変数は大域的なものとなるので (要するに、どこからでも見える)、名前の衝突[9]に気を付ける必要がある。例えば

```
PowerSum[x_, n_] := Sum[x^i, {i,1,n}]
```

とすると、“PowerSum[x,5]” のようなのは大丈夫だが、“PowerSum[i,5]” はダメになる。“Module[{*local-var1,local-var2,..*}, *procedure*]” を利用して

```
PowerSum[x_,n_] :=
  Module[{i},              ← 局所変数 i を使うことを宣言
    Sum[x^i, {i,1,n}]
  ]
```

とする方が良い (Module[] に似たものに Block[] という関数がある。前者は変数そのものを局所的に作るが、後者は変数の値だけを局所的に作る。)。

context の利用 実は、上の例はまだ完璧とは言えない。“i” という名前そのものはグローバルに見えてしまう (あまり実害はないけれど)。context (文脈と訳されることが多い) を導入して、名前の扱いを制御する。

```
Begin["Private`"]
  PowerSum[x_,n_] :=
    Module[{i},
      Sum[x^i, {i,1,n}]
    ]
End[]
```

この例では “Private`” という context を導入している。とはいえ、他人に使ってもらうプログラムを書く場合でもなければ、ここまで気にする必要はないかもしれない。

例 1.

先に出した Fibonacci 数列を計算する関数はあまり効率が良くないので、Do[] を用いて書いてみる:

```
Fibonacci[n_Integer?Positive] :=
  Module[{fn=1,fn2=0},
    Do[{fn1,fn2}={fn1+fn2,fn1}, {n-1}];
    fn1
  ]
```

例 2.

平均値、分散を計算するプログラム (関数) を作れ。

[9]異なるものに同じ名前をつけようとすると、しかられたり、古い方が上書きされて、動作が変になったりする。

```
mean::usage =
  "mean[list] returns the mean value of the elements of list."
variation::usage =
  "variation[list] returns the variation of the elements of list."

Begin["Private`"]
mean[l_List] :=
  Module[{n = Length[l], i}, Sum[l[[i]],{i,n}] / n]
variation[l_List] :=
  Module[{n, m, i},
        n = Length[l]; m=mean[l]; Sum[(l[[i]]-m)^2, {i,n}]/n]
End[]
```

Mathematica では C のように何かをするための専用プログラムを記述して、コンパイルして実行するのではなく、知識を増やして賢くして行き、質問に答えさせるという感じ。何か複雑なことを計算させるときは、○○関係の関数集 (パッケージという) をシステムにロードして使う、という形態を取る。こういう場合、名前をきちんと扱うのは重要になる。(人間の知的活動でも、一つの名前が、状況によって全く違う意味になることがあるが、文脈の違いを適当に認識することによって使い分けている。)

7 Mathematica のグラフィックス

7.1 グラフィックス用の代表的な関数

まず、どんな関数が用意されているかざっと見てみよう。

2 次元グラフ Plot[] (関数のグラフ), ParametricPlot[] (パラメーター曲線の描画), ListPlot[] (リスト・データのプロット)

3 次元グラフ Plot3D[] (2 変数関数のグラフ), ParametricPlot3D[] (パラメーター曲線・曲面の描画), ListPlot3D[]

等高線グラフ ContourPlot[] (2 変数関数の等高線), ListContourPlot[]

濃淡グラフ DensityPlot[] (2 変数関数の濃淡図), ListDensityPlot[]

7.2 グラフィックス・オブジェクト

グラフィックスの関数では、対応するグラフィックス・オブジェクトを生成して、画面に表示する。グラフィックス・オブジェクトは変数に代入可能である。オブジェクトを表示するには関数 Show[] を用いる。またオブジェクトがどう表現されているかは InputForm[] で読むことが出来る。

次の例では、変数 g1, g2 に代入しておいて、後から再表示している。

```
g1 = Plot[Sin[x],{x,0,2Pi}]
g2 = Plot[Cos[x],{x,0,2Pi}]
Show[g1]
Show[g1,g2]
InputForm[g1]
```

7.3 グラフィックスの印刷のしかた

関数 Export[] を “Export["ファイル名 "], グラフィックス]” のように使って、グラフィックスをファイルに出力できる。ファイル名の末尾を .eps にしておくと、自動的に EPS (Encapsulated PostScript) フォーマットが選択されて便利である。

その他に .jpg, .gif などの画像フォーマットも利用できる。

トーラスの図をファイルに記録

```
g = ParametricPlot3D[{Cos[t](3+Cos[u]),Sin[t](3+Cos[u]),Sin[u]},
           {t,0,2Pi}, {u,0,2Pi}]
Export["torus.eps", g]
```

とすると “torus.eps” というファイルが出来るが、これを印刷するには

torus.eps を印刷

```
waltz21% lp -d プリンタ名 torus.eps
```

のようにしてプリンターに送ればよい。

ファイルの名前を末尾を “.eps” にすることで、出力の形式として EPS (Encapsulated PostScript) を採用してくれる。こうして作成した EPS ファイルは LaTeX 文書に取り込むのも簡単である。

大昔はこうだった (古いバージョンを使うときの参考まで)

Display[] 関数を “ Display["!psfix > ファイル名", グラフィックス]” のように使うと図形を印刷するための PostScript データが作成できる。

例えば

トーラスの図をファイルに記録

```
g = ParametricPlot3D[{Cos[t](3+Cos[u]),Sin[t](3+Cos[u]),Sin[u]},
      {t,0,2Pi}, {u,0,2Pi}]
Display["!psfix > torus.ps", g]
```

7.4 グラフィックスの関数の引数、特にオプション

グラフィックスの関数が、引数として取るものは、順に

(1) 関数あるいは関数のリスト or 数値データによるリスト

(2) 描画する範囲

(3) 描画を制御するオプション

このうちオプションはその名前の通り省略可能である。

(オプションとは？実際にどうやって描画するかについて、自由度があり、それを一々指定するのはとても面倒なので、普段はそれを適当に決めているが、そのデフォールト値で満足できない場合に、ユーザーが自分の求めるものを指定することが出来るようにしたもの。)

関数がどういうオプションを持つかは “??関数名” で調べられる。例えば、関数 Plot[] のオプションを調べたければ、次のようにすればよい。

```
??Plot
```

オプションの指定の仕方は、” オプション名 -> 値” である。例えば

```
AspectRatio -> 数値              デフォールト値は 1/GoldenRatio
                                 Automatic とすると 1:1
Axes -> 真偽値
AxesLabel -> {"x", "y=f(x)"}
Compiled -> False                デフォールト値は True
Frame -> True                    デフォールト値は False
GridLines -> Automatic
PlotRange -> {zmin,zmax}
PlotPoints->100
```

しばらくの間、オプションを変更したまま使いたい場合は、SetOptions を使う。次の例では、関数 Plot3D 使用時に PlotPoints を 100 にするよう設定している。

```
SetOptions[Plot3D, PlotPoints->100]
```

7.5 2次元グラフィックス

7.5.1 Plot[]

1 変数関数のグラフを描くには Plot[] を使う。基本的な使い方は

```
Plot[関数，描画範囲を示すリスト]
Plot[{関数 1, 関数 2, ..., 関数 n}，描画範囲を示すリスト]
```

描画範囲を示すリストから、関数値を計算するための変数の適当な値を選びだして、そこでの関数値を評価して、そうして得た点 (標本点) を順に結んでグラフを描く。標本点の個数を多くするには PlotPoints オプションを指定する。例として

```
Plot[Sin[x], {x,0,2Pi}]
Plot[Sin[x], Sin[2x], Sin[3x], {x,0,2Pi}]
Plot[Sin[50x],{x,0,Pi},PlotPoints->1000]
```

7.5.2 ParametricPlot[]

いわゆるパラメーター表示された平面曲線を描く。例えば

```
ParametricPlot[{Sin[t], Sin[2t]}, {t,0,2Pi}]
```

7.5.3 ListPlot[]

数値データのリストから曲線を描く。これを用いると、例えば外部のプログラムで計算したデータを表示できる。次の例では Table[] で生成したリストから曲線を描いている。

```
fp = Table[{t,N[Sin[Pi t]]}, {t,0,0.5,0.025}]
ListPlot[fp]
ListPlot[fp, PlotJoined -> True]
```

7.5.4 DSolve[], NDSolve[]

```
DSolve[x^3 y'[x]+y[x]^2==0,y[x],x]
NDSolve[{y'[x]==Sin[y[x]],y[0]==1}, y, {x,0,4}]
Plot[Evaluate[y[x] /.  %], {x,0,4}]
```

(この NDSolve[] は数値的に微分方程式を解くコマンドであり、結果は離散的な点での関数値を近似的に求めたものである。計算していない点での値は補間により簡略計算する。こういうものを InterpolatingFunction という。)

```
y[1.5] /. %%
```

7.5.5 2 変数関数の等高線、濃淡図

2 変数関数 $(x, y) \mapsto f(x, y)$ を可視化するために、等高線を描く ContourPlot[][10], 濃淡図を描く DensityPlot[] が用意されている。 使い方は

```
ContourPlot[f, {x,xmin,xmax}, {y,ymin,ymax}]
DensityPlot[f, {x,xmin,xmax}, {y,ymin,ymax}]
```

[10]等高線のことを contour (line) と呼ぶ。

例として

```
ContourPlot[Sin[x]Sin[y], {x,-2,2}, {y,-2,2}]
```

オプションとして

Contours -> 整数
PlotRange -> {zmin,zmax} または Automatic
PlotPoints -> 整数 （デフォールトが 15。小さい！非力さが出て来る。）
ContourShading -> False （デフォールトは True）

2 変数関数についての方程式で定義される陰関数のグラフを描くために、ImplicitPlot[] という関数があるが (ただし古いバージョンの Mathematica にはない)、それを使うには事前に Needs["Graphics`ImplicitPlot`"] を実行する必要がある。

```
Needs["Graphics`ImplicitPlot`"]
ImplicitPlot[x^2+y^2==1, {x,-1,1}]
```

7.6 ３次元グラフィックス

2 変数関数のグラフを描く Plot3D[] は

Plot3D[f, {x,xmin,xmax}, {y,ymin,ymax}]

が基本的な使い方。例えば

```
Plot3D[Sin[x y], {x,0,3}, {y,0,3}]
```

オプションとして以下のようなものがある。

HiddenSurface -> False	隠面消去をしない
PlotPoints -> 個数	大きくすると細かい図を書く。
ViewPoint -> {x,y,z}	視点の指定

ParametricPlot3D[], ListPlot3D[] 等もある。次の例は是非試してみよう。

```
ParametricPlot3D[{Sin[t], Cos[t], t/3}, {t, 0, 15}]
ParametricPlot3D[{t,u,Sin[t u]}, {t,0,3}, {u,0,3}]
ParametricPlot3D[{Sin[t],Cos[t],u}, {t,0,2Pi}, {u,0,4}]
ParametricPlot3D[{Cos[t](3+Cos[u]),Sin[t](3+Cos[u]),Sin[u]},
        {t,0,2Pi}, {u,0,2Pi}]
ParametricPlot3D[{Cos[t]Cos[u],Sin[t]Cos[u],Sin[u]},
        {t,0,2Pi}, {u,-Pi/2,Pi/2}]
```

8　Mathematica のサウンド機能

(準備中)

9　ある年度の情報処理 II レポート問題

問題 1

高校での数学の計算問題、大学の微分積分や線形代数の計算問題を解かせてみる。どこまで解けるか、どういう問題が解けないか (ごくまれに結果がおかしいこともある)、色々試し、また解けないのは何故か推測する。(問題によっては工夫が必要かもしれない。簡単に Mathematica では解けないと決めつけないで、考えること。色々考えた経過をレポートする。)

問題 2

1 変数、または 2 変数の関数 f を定め、グラフ、等高線、微分が 0 になる点、またその点での関数値等を計算する。多変数関数の極値問題を解く際にどのように利用できるか、試して見よ。(この方向に発展させると、極値問題を解くプログラムが作れるかも知れない? これまで完全なプログラムを書いた学生はいない。挑戦を期待する。

10　工事計画

- ノートブックの使い方
- 画像の印刷など
- 音声の取り扱い
- パレット
- レポートの作り方

A　式の評価順序の話 — Evaluate[]

```
Table[BesselJ[n,x],{n,5}]
```

とすると、{BesselJ[1,x],BesselJ[2,x],BesselJ[3,x],BesselJ[4,x],BesselJ[5,x]} という 5 つの関数を含んだリストができる。一方、

```
Plot[{BesselJ[1,x],BesselJ[2,x],BesselJ[3,x],BesselJ[4,x],
BesselJ[5,x]},{x,0.0,10.0}]
```

とすると、その 5 つの関数のグラフが表示される。すると、

```
Plot[Table[BesselJ[n,x],n,5],{x,0.0,10.0}]
```

とすれば 5 つの関数のグラフが描けるような気がするが、実はうまく行かない。Lisper だったら、その理由は分かるでしょう。そして、

```
Plot[Evaluate[Table[BesselJ[n,x],{n,5}]],{x,0,10}]
```

とすると、うまく行くというのにも納得できるでしょう。一件落着。

同様に

```
f[x_]:=Sin[x]
```

としたとき、D[f[x],x] は Cos[x] となるが、

```
Plot[D[f[x],x],{x,0,2Pi}]
```

としても Cos[x] のグラフは描けない。

```
Plot[Evaluate[D[f[x],x]],{x,0,2Pi}]
```

あるいは、

```
Plot[f'[x],{x,0,2Pi}]
```

とする。

B プログラム例

B.1 2 次元の Newton 法

```
f[x_,y_]:={x^2-y^2+x+1,2 x y +y}
Df[a_,b_]:=Module[
            {x,y},
            Transpose[{D[f[x,y],x],D[f[x,y],y]}] /. {x->a,y->b}
           ]
{x,y}={1,1}
Do[{x,y}={x,y}-Inverse[Df[x,y]].f[x,y];
   Print[{x,y},"=",N[{p,q},20]],
   {6}
  ]
```

あるいは、変数の方もベクトル的にリストを使って書いて、

```
f[{x_,y_}]:={x^2-y^2+x+1,2 x y +y}
Df[{a_,b_}]:=Module[
            {x,y},
            Transpose[{D[f[{x,y}],x],D[f[{x,y}],y]}] /. {x->a,y->b}
           ]
xk={1,1}
Do[xk=xk-Inverse[Df[xk]].f[xk]; Print[xk,"=",N[xk,20]], {6}]
```

B.2 二変数関数の極値問題 kyokuchi.m

```
(* 二変数関数 f の停留点を求める（よう努力する）*)
teiryuuten[f_]:=
Module[
  {fx,fy},
  fx=Simplify[D[f[x,y],x]];
  fy=Simplify[D[f[x,y],y]];
  Solve[{fx==0,fy==0},{x,y}]
]

(* 二変数関数 f とその停留点のリスト s を分析し、極値の判定をする *)
bunseki[s_,f_]:=
Module[
  {ff,HesseXY,aSolution,restSolutions,valf,l1,l2},
  ff=f[x,y];
  HesseXY = {{D[ff,x,x],D[ff,x,y]},
             {D[ff,y,x],D[ff,y,y]}};
  restSolutions = s;
  While [(restSolutions != {}),
         aSolution = First[restSolutions];
         restSolutions = Rest[restSolutions];
         valf = ff /. aSolution;
         {l1,l2} = Eigenvalues[HesseXY /. aSolution];
         If [l1 > 0 && l2 > 0,
            Print[aSolution, ", 極小 f(x,y)=", valf]];
         If [l1 < 0 && l2 < 0,
            Print[aSolution, ", 極大 f(x,y)=", valf]];
         If [(l1 l2 < 0),
            Print[aSolution, ", 極値でない"]];
         If [(l1 l2 == 0),
            Print[aSolution, ", 極値であるかどうか分からない。"]];
      ]
]
```

使用例

```
oyabun% math
Mathematica 4.0 for Solaris
Copyright 1988-1999 Wolfram Research, Inc.
 -- Motif graphics initialized --

In[1]:= << /home/syori2/kyokuchi.m

In[2]:= f[x_,y_]:=x y(x^2+y^2-4)

In[3]:= s=teiryuuten[f]

Out[3]= {{x -> -2, y -> 0}, {x -> -1, y -> -1}, {x -> -1, y -> 1},

>    {x -> 0, y -> 0}, {x -> 1, y -> -1}, {x -> 1, y -> 1}, {x -> 2, y -> 0},

>    {y -> -2, x -> 0}, {y -> 2, x -> 0}}

In[4]:= bunseki[s,f]
{x -> -2, y -> 0}, 極値でない
{x -> -1, y -> -1}, 極小 f(x,y)=-2
{x -> -1, y -> 1}, 極大 f(x,y)=2
{x -> 0, y -> 0}, 極値でない
{x -> 1, y -> -1}, 極大 f(x,y)=2
{x -> 1, y -> 1}, 極小 f(x,y)=-2
{x -> 2, y -> 0}, 極値でない
{y -> -2, x -> 0}, 極値でない
{y -> 2, x -> 0}, 極値でない

In[5]:=
```